NAG Library Routine Document

G05RKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G05RKF generates pseudorandom uniform variates with joint distribution of a Gumbel–Hougaard Archimedean copula.

2 Specification

```
SUBROUTINE G05RKF (N, M, THETA, SORDER, STATE, X, LDX, SDX, IFAIL)
INTEGER              N, M, SORDER, STATE(*), LDX, SDX, IFAIL
REAL (KIND=nag_wp) THETA, X(LDX,SDX)
```

3 Description

Generates n pseudorandom uniform m-variates whose joint distribution is the Gumbel–Hougaard Archimedean copula C_θ, given by

$$C_\theta = \exp\left\{-\left[(-\ln u_1)^\theta + (-\ln u_2)^\theta + \cdots + (-\ln u_m)^\theta\right]\right\}, \quad \begin{cases} \theta \in (1,\infty), \\ u_j \in (0,1], \quad j = 1,2,\ldots m; \end{cases}$$

with the special cases:

$C_1 = u_1 u_2 \cdots u_m$, the product copula;

$C_\infty = \min(u_1, u_2, \ldots, u_m)$, the Fréchet–Hoeffding upper bound.

The generation method uses mixture of powers.

One of the initialization routines G05KFF (for a repeatable sequence if computed sequentially) or G05KGF (for a non-repeatable sequence) must be called prior to the first call to G05RKF.

4 References

Marshall A W and Olkin I (1988) Families of multivariate distributions *Journal of the American Statistical Association* **83** 403

Nelsen R B (2006) *An Introduction to Copulas* (2nd Edition) Springer Series in Statistics

5 Arguments

1: N – INTEGER *Input*

On entry: n, the number of pseudorandom uniform variates to generate.

Constraint: $N \geq 0$.

2: M – INTEGER *Input*

On entry: m, the number of dimensions.

Constraint: $M \geq 2$.

3: THETA – REAL (KIND=nag_wp) *Input*

On entry: θ, the copula parameter.

Constraint: THETA ≥ 1.0.

4: SORDER – INTEGER *Input*

On entry: determines the storage order of variates; the (i, j)th variate is stored in $\mathrm{X}(i, j)$ if SORDER $= 1$, and $\mathrm{X}(j, i)$ if SORDER $= 2$, for $i = 1, 2, \ldots, n$ and $j = 1, 2, \ldots, m$.

Constraint: SORDER $= 1$ or 2.

5: STATE($*$) – INTEGER array *Communication Array*

Note: the actual argument supplied **must** be the array STATE supplied to the initialization routines G05KFF or G05KGF.

On entry: contains information on the selected base generator and its current state.

On exit: contains updated information on the state of the generator.

6: X(LDX, SDX) – REAL (KIND=nag_wp) array *Output*

On exit: the pseudorandom uniform variates with joint distribution described by C_θ, with $\mathrm{X}(i, j)$ holding the ith value for the jth dimension if SORDER $= 1$ and the jth value for the ith dimension of SORDER $= 2$.

7: LDX – INTEGER *Input*

On entry: the first dimension of the array X as declared in the (sub)program from which G05RKF is called.

Constraints:

if SORDER $= 1$, LDX $\geq$ N;
if SORDER $= 2$, LDX $\geq$ M.

8: SDX – INTEGER *Input*

On entry: the second dimension of the array X as declared in the (sub)program from which G05RKF is called.

Constraints:

if SORDER $= 1$, SDX $\geq$ M;
if SORDER $= 2$, SDX $\geq$ N.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= 1$

On entry, corrupt STATE argument.

IFAIL $= 2$

On entry, invalid THETA: THETA $= \langle value \rangle$.
Constraint: THETA ≥ 1.0.

IFAIL $= 3$

On entry, N $= \langle value \rangle$.
Constraint: N ≥ 0.

IFAIL $= 4$

On entry, M $= \langle value \rangle$.
Constraint: M ≥ 2.

IFAIL $= 5$

On entry, invalid SORDER.
Constraint: SORDER $= 1$ or 2.

IFAIL $= 7$

On entry, LDX must be at least $\langle value \rangle$: LDX $= \langle value \rangle$.

IFAIL $= 8$

On entry, SDX must be at least $\langle value \rangle$: SDX $= \langle value \rangle$.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

Not applicable.

8 Parallelism and Performance

G05RKF is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

In practice, the need for numerical stability restricts the range of θ such that:

> if $(\theta - 1) < 1.0 \times 10^{-6}$, the routine returns pseudorandom uniform variates with C_1 joint distribution;
>
> if $\theta > \max(80.0, -0.5 \ln \epsilon_s)$, the routine returns pseudorandom uniform variates with C_∞ joint distribution;

where ϵ_s is the safe-range parameter, the value of which is returned by X02AMF.

10 Example

This example generates thirteen four-dimensional variates for copula $C_{2.4}$.

10.1 Program Text

```
    Program g05rkfe

!     G05RKF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g05kff, g05rkf, nag_wp, x04caf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter                 :: lseed = 1, nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)                 :: theta
      Integer                            :: genid, ifail, ldx, lstate, m, n,     &
                                            sdx, sorder, subid
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable    :: x(:,:)
      Integer                            :: seed(lseed)
      Integer, Allocatable               :: state(:)
!     .. Executable Statements ..
      Write (nout,*) 'G05RKF Example Program Results'
      Write (nout,*)
      Flush (nout)

!     Skip heading in data file
      Read (nin,*)

!     Read in the base generator information and seed
      Read (nin,*) genid, subid, seed(1)

!     Initial call to initializer to get size of STATE array
      lstate = 0
      Allocate (state(lstate))
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)

!     Reallocate STATE
      Deallocate (state)
      Allocate (state(lstate))

!     Initialize the generator to a repeatable sequence
      ifail = 0
      Call g05kff(genid,subid,seed,lseed,state,lstate,ifail)
```

```
!     Read in sample size, number of dimensions and order
      Read (nin,*) n, m, sorder

      If (sorder==1) Then
!       X(N,M)
        ldx = n
        sdx = m
      Else
!       X(M,N)
        ldx = m
        sdx = n
      End If
      Allocate (x(ldx,sdx))

!     Read in parameter
      Read (nin,*) theta

!     Generate variates
      ifail = 0
      Call g05rkf(n,m,theta,sorder,state,x,ldx,sdx,ifail)

!     Display the variates
      If (sorder==1) Then
!       X(N,M)
        ifail = 0
        Call x04caf('General',' ',n,m,x,ldx,                                   &
          'Uniform variates with copula joint distribution',ifail)
      Else
!       X(M,N)
        ifail = 0
        Call x04caf('General',' ',m,n,x,ldx,                                   &
          'Uniform variates with copula joint distribution',ifail)
      End If

    End Program g05rkfe
```

10.2 Program Data

```
G05RKF Example Program Data
1  1  1762543          :: GENID,SUBID,SEED(1)
13  4  1               :: N,M,SORDER
2.4                    :: THETA
```

10.3 Program Results

```
 G05RKF Example Program Results

 Uniform variates with copula joint distribution
          1       2       3       4
  1   0.9369  0.8676  0.9713  0.8854
  2   0.1139  0.3063  0.8625  0.2743
  3   0.4418  0.2211  0.5042  0.4985
  4   0.7902  0.6007  0.7493  0.6474
  5   0.8362  0.9847  0.8807  0.9079
  6   0.1781  0.4610  0.1283  0.1329
  7   0.1272  0.1760  0.1805  0.0383
  8   0.4473  0.2171  0.1662  0.1300
  9   0.8899  0.9005  0.8844  0.8879
 10   0.9069  0.8681  0.8450  0.8804
 11   0.2222  0.5499  0.4965  0.6488
 12   0.3807  0.5967  0.5096  0.3577
 13   0.8445  0.7755  0.8661  0.8948
```